Variations of Metric With xTras

Downloading and Installing xAct

Visit the page: http://www.xact.es
Follow installation instructions on http://www.xact.es/download.html
Linux:
1. Download the tarball xAct_V.tgz (V is the version number)
2. sudo -i ; cd /usr/share/Mathematica/Applications/; tar xvfz ~/Downloads/xAct_V.tgz
Windows:
1. Download the zip file xAct_V.zip (V is the version number)
2. unzip its contents in C:\Program Files\Wolfram Research\Mathematica\<version>\AddOns\Applications\

Read the documentation:
http://www.xact.es/documentation.html

The documentation is also installed locally, after you load the package, run the command:
xTrasHelp[]

most likely in: (look for the files xTras.pdf, MetricVariations.nb using your file searching tool)
Linux: /usr/share/Mathematica/Applications/xAct/xTras/Documentation/English/
See: /usr/share/Mathematica/Applications/xAct/xTras/Documentation/English/**xTras.pdf**
/usr/share/Mathematica/Applications/xAct/xTras/Documentation/English/Guides/**xTras.nb**

/usr/share/Mathematica/Applications/xAct/xTras/Documentation/English/Tutorials/**MetricVariations.nb**

Metric Variations

xTras automatically defines 'proper' variations with respect to the metric via `VarD` and `VarL` whenever you define a metric.

VarD[g[−a, −b], cd][L] returns $\frac{\delta L}{\delta g_{ab}}$ while integrating by parts with respec
covariant derivative cd.

VarL[g[−a, −b], cd][L] returns $\frac{1}{\sqrt{-g}} \frac{\delta\left(\sqrt{-g}\; L\right)}{\delta g_{ab}}$ while integrating by parts with re
covariant derivative cd.

First start a session:

In[ ]:=
```
Needs["xAct`xTras`"]
```

In[ ]:=
```
DefManifold[M4, 4, {λ, μ, ν, ρ, σ, α, β, γ, δ}];
DefMetric[-1, g[-μ, -ν], CD, PrintAs → "g", CurvatureRelations → True];
```

```
** DefManifold: Defining manifold M4.
** DefVBundle: Defining vbundle TangentM4.
** DefTensor: Defining symmetric metric tensor g[-μ, -ν].
** DefTensor: Defining antisymmetric tensor epsilong[-α, -β, -γ, -δ].
** DefTensor: Defining tetrametric Tetrag[-α, -β, -γ, -δ].
** DefTensor: Defining tetrametric Tetrag†[-α, -β, -γ, -δ].
** DefCovD: Defining covariant derivative CD[-μ].
** DefTensor: Defining vanishing torsion tensor TorsionCD[α, -β, -γ].
** DefTensor: Defining symmetric Christoffel tensor ChristoffelCD[α, -β, -γ].
** DefTensor: Defining Riemann tensor RiemannCD[-α, -β, -γ, -δ].
** DefTensor: Defining symmetric Ricci tensor RicciCD[-α, -β].
** DefCovD:  Contractions of Riemann automatically replaced by Ricci.
** DefTensor: Defining Ricci scalar RicciScalarCD[].
** DefCovD:  Contractions of Ricci automatically replaced by RicciScalar.
** DefTensor: Defining symmetric Einstein tensor EinsteinCD[-α, -β].
** DefTensor: Defining Weyl tensor WeylCD[-α, -β, -γ, -δ].
** DefTensor: Defining symmetric TFRicci tensor TFRicciCD[-α, -β].
** DefTensor: Defining Kretschmann scalar KretschmannCD[].
** DefCovD:  Computing RiemannToWeylRules for dim 4
** DefCovD:  Computing RicciToTFRicci for dim 4
** DefCovD:  Computing RicciToEinsteinRules for dim 4
** DefTensor: Defining symmetrized Riemann tensor SymRiemannCD[-α, -β, -γ, -δ].
** DefTensor: Defining symmetric Schouten tensor SchoutenCD[-α, -β].
** DefTensor: Defining symmetric cosmological Schouten tensor SchoutenCCCD[LI[_], -α, -β].
** DefTensor: Defining symmetric cosmological Einstein tensor EinsteinCCCD[LI[_], -α, -β].
** DefTensor: Defining weight +2 density Detg[]. Determinant.
** DefParameter: Defining parameter PerturbationParameterg.
** DefTensor: Defining tensor Perturbationg[LI[order], -α, -β].
```

Get help:

```
xTrasHelp[]; (* opens a window with a notebook to browse for help *)
```

You can perform variations with respect to $\delta g^{\mu\nu}$ or $\delta g_{\mu\nu}$:

Variations are expected to be performed on scalar/scalar density expressions

In[ ]:=
```
Print[
  "δ/δg^μν g= ", VarD[g[ μ,  ν], CD][Detg[]] , "\n",
  "δ/δg_μν g= ", VarD[g[-μ, -ν], CD][Detg[]] , "\n",
  "δ/δg^μν √g = ", VarD[g[ μ,  ν], CD][√(-Detg[])]
]
```

$$\frac{\delta}{\delta g^{\mu\nu}} g = -\tilde{\tilde{g}}\; g_{\mu\nu}$$

$$\frac{\delta}{\delta g_{\mu\nu}} g = \tilde{\tilde{g}}\; g^{\mu\nu}$$

$$\frac{\delta}{\delta g^{\mu\nu}} \sqrt{g} = -\frac{1}{2} \sqrt{-\tilde{\tilde{g}}}\; g_{\mu\nu}$$

VarL simply takes care the $\sqrt{-g}$ factors in integrals:

In[ ]:=
```
Print[
  "1/√-g δ/δg^μν √-g = ", VarL[g[ μ,  ν], CD][1] , "\n",
  "1/√-g δ/δg_μν √-g = ", VarL[g[-μ, -ν], CD][1]
]
```

$$\frac{1}{\sqrt{-g}} \frac{\delta}{\delta g^{\mu\nu}} \sqrt{-g} = -\frac{1}{2}\; g_{\mu\nu}$$

$$\frac{1}{\sqrt{-g}} \frac{\delta}{\delta g_{\mu\nu}} \sqrt{-g} = \frac{g^{\mu\nu}}{2}$$

Einstein-Hilbert action:

In[ ]:=
```
Print[
 "δ/δg^μν R= ", VarD[g[ μ,  ν], CD][RicciScalarCD[]], "\n",
 "1/√-g δ/δg^μν (√-g  R)= ",
 1/√-Detg[] VarD[g[ μ,  ν], CD][√-Detg[] RicciScalarCD[]] // Simplification, "\n",
 "1/√-g δ/δg^μν (√-g  R)= ", VarL[g[ μ,  ν], CD][RicciScalarCD[]], "\n"
]
```

$$\frac{\delta}{\delta g^{\mu\nu}} R= R[\nabla]_{\mu\nu}$$

$$\frac{1}{\sqrt{-g}}\frac{\delta}{\delta g^{\mu\nu}}(\sqrt{-g}\ R)= R[\nabla]_{\mu\nu} - \frac{1}{2} g_{\mu\nu} R[\nabla]$$

$$\frac{1}{\sqrt{-g}}\frac{\delta}{\delta g^{\mu\nu}}(\sqrt{-g}\ R)= R[\nabla]_{\mu\nu} - \frac{1}{2} g_{\mu\nu} R[\nabla]$$

We can vary more complicated expressions:

In[ ]:=
```
Print[
 "1/√-g δ/δg^μν (√-g  R_αβ R^αβ)= ",
 VarL[g[ μ,  ν], CD][RicciCD[-α, -β] RicciCD[α, β]] // ContractMetric // ToCanonical //
  ScreenDollarIndices
]
```

$$\frac{1}{\sqrt{-g}}\frac{\delta}{\delta g^{\mu\nu}}(\sqrt{-g}\ R_{\alpha\beta}R^{\alpha\beta})=$$

$$-\frac{1}{2} g_{\mu\nu} R[\nabla]_{\alpha\beta} R[\nabla]^{\alpha\beta} + 2 R[\nabla]_{\mu\alpha} R[\nabla]_{\nu}{}^{\alpha} - \nabla_{\alpha}\nabla_{\mu} R[\nabla]_{\nu}{}^{\alpha} - \nabla_{\alpha}\nabla_{\nu} R[\nabla]_{\mu}{}^{\alpha} + \nabla^{\alpha}\nabla_{\alpha} R[\nabla]_{\mu\nu} + g_{\mu\nu}\left(\nabla_{\beta}\nabla_{\alpha} R[\nabla]^{\alpha\beta}\right)$$

FullSimplification[ ] does a better job: Takes Bianchi identities into account, as well as sorting derivatives.

In[ ]:=
```
Print[
 "1/√-g δ/δg^μν (√-g RαβR^αβ)= ",
 VarL[g[ μ,  ν], CD][RicciCD[-α, -β] RicciCD[α, β]] // FullSimplification[] //
  ScreenDollarIndices
]
```

$$\frac{1}{\sqrt{-g}}\frac{\delta}{\delta g^{\mu\nu}}(\sqrt{-g}\ R_{\alpha\beta}R^{\alpha\beta})=$$

$$-\frac{1}{2}\ g_{\mu\nu}\ \ R[\nabla]_{\alpha\beta}\ \ R[\nabla]^{\alpha\beta}+2\ R[\nabla]^{\alpha\beta}\ \ R[\nabla]_{\mu\alpha\nu\beta}+\nabla_\alpha\nabla^\alpha R[\nabla]_{\mu\nu}+\frac{1}{2}\ g_{\mu\nu}\ \ (\nabla_\alpha\nabla^\alpha R[\nabla])-\nabla_\nu\nabla_\mu R[\nabla]$$

In[ ]:=
```
Print[
 "1/√-g δ/δg^μν (√-g RαβγδR^αβγδ)= ",
 VarL[g[ μ,  ν], CD][KretschmannCD[]] // FullSimplification[] // ScreenDollarIndices
]
```

$$\frac{1}{\sqrt{-g}}\frac{\delta}{\delta g^{\mu\nu}}(\sqrt{-g}\ R_{\alpha\beta\gamma\delta}R^{\alpha\beta\gamma\delta})=$$

$$-\frac{1}{2}\ g_{\mu\nu}\ K[\nabla]-4\ R[\nabla]_\mu{}^\alpha\ \ R[\nabla]_{\nu\alpha}+4\ R[\nabla]^{\alpha\beta}\ \ R[\nabla]_{\mu\alpha\nu\beta}+2\ R[\nabla]_\mu{}^{\alpha\beta\gamma}\ \ R[\nabla]_{\nu\alpha\beta\gamma}+4\left(\nabla_\alpha\nabla^\alpha R[\nabla]_{\mu\nu}\right)-2\ (\nabla_\nu\nabla_\mu R[\nabla])$$

Scalar Fields

In[ ]:=
```
DefTensor[ϕ[], M4, PrintAs → "ϕ"];
DefConstantSymbol[{Λ, κ, c}];
```

```
** DefConstantSymbol: Defining constant symbol Λ.
** DefConstantSymbol: Defining constant symbol κ.
** DefConstantSymbol: Defining constant symbol c.
```

Vary the free real scalar field Lagrangian to obtain the energy momentum tensor:

In[ ]:=
```
Print[
 "T_{μν}= -2/Sqrt[-g] δ/δg^{μν} ( Sqrt[-g] ℒ)= ",
 -2 VarL[g[ μ,  ν], CD][-1/2 (CD[-μ]@ ϕ[]) (CD[μ]@ ϕ[]) - 1/2 ϕ[]^2] // Simplify //
  ScreenDollarIndices
]
```

$$T_{\mu\nu}= -\frac{2}{\sqrt{-g}}\frac{\delta}{\delta g^{\mu\nu}}(\sqrt{-g}\,\mathcal{L})= -\frac{1}{2}\; g_{\mu\nu}\;\left(\phi^2+(\nabla_\alpha \phi)(\nabla^\alpha \phi)\right)+(\nabla_\mu \phi)(\nabla_\nu \phi)$$

Vary with respect to ϕ to obtain equations of motion:

In[ ]:=
```
Print[
 "δ/δϕ ℒ = ", VarD[ϕ[], PD][-1/2 (PD[-μ]@ ϕ[]) (PD[μ]@ ϕ[]) - 1/2 ϕ[]^2] // FullSimplification[] //
  ScreenDollarIndices, "  (flat spacetime)\n",
 "      = ", VarD[ϕ[], CD][-1/2 (CD[-μ]@ ϕ[]) (CD[μ]@ ϕ[]) - 1/2 ϕ[]^2] // FullSimplification[] //
  ScreenDollarIndices, "\n",
 "      = ", VarD[ϕ[], CD][+1/2 ϕ[] (CD[-μ]@CD[μ]@ ϕ[]) - 1/2 ϕ[]^2] // FullSimplification[] //
  ScreenDollarIndices
]
```

$$\frac{\delta}{\delta\phi}\mathcal{L}\;=\;-\phi+\frac{\partial_\alpha\partial^\alpha\phi}{2}+\frac{\partial^\alpha\partial_\alpha\phi}{2}\quad \text{(flat spacetime)}$$

$$=\;-\phi+\nabla_\alpha\nabla^\alpha\,\phi$$

$$=\;-\phi+\nabla_\alpha\nabla^\alpha\,\phi$$

Other terms, present in other theories:

In[ ]:=
```
Print[
 "1/Sqrt[-g] δ/δg^{μν} ( Sqrt[-g]  e^ϕ  R)= ",
 VarL[g[ μ,  ν], CD][e^ϕ[] RicciScalarCD[]] // FullSimplification[] // ScreenDollarIndices
]
```

$$\frac{1}{\sqrt{-g}}\frac{\delta}{\delta g^{\mu\nu}}(\sqrt{-g}\;\; e^\phi\;\; R)=$$

$$e^\phi\;R[\nabla]_{\mu\nu}-\frac{1}{2}\,e^\phi\;g_{\mu\nu}\;R[\nabla]+e^\phi\;g_{\mu\nu}\;(\nabla_\alpha\nabla^\alpha\phi)+e^\phi\;g_{\mu\nu}\;(\nabla_\alpha\phi)(\nabla^\alpha\phi)-e^\phi(\nabla_\mu\phi)(\nabla_\nu\phi)-e^\phi(\nabla_\nu\nabla_\mu\phi)$$

In[ ]:=
```
Print[
 "1/√-g δ/δg^μν ( √-g  G_μν∇^μϕ ∇^νϕ)= ",
 VarL[g[ μ,  ν], CD][EinsteinCD[-μ, -ν] (CD[μ]@ϕ[]) (CD[ν]@ϕ[])] // FullSimplification[] //
  ScreenDollarIndices
]
```

$$\frac{1}{\sqrt{-g}}\frac{\delta}{\delta g^{\mu\nu}}(\sqrt{-g}\;\; G_{\mu\nu}\nabla^{\mu}\phi\;\;\nabla^{\nu}\phi)= -\frac{1}{2}\;R[\nabla]_{\mu\nu}\;(\nabla_{\alpha}\phi)(\nabla^{\alpha}\phi)+\frac{1}{2}\;g_{\mu\nu}\;(\nabla_{\alpha}\nabla^{\alpha}\phi)(\nabla_{\beta}\nabla^{\beta}\phi)-\frac{1}{2}\;G[\nabla]_{\alpha\beta}\;g_{\mu\nu}\;(\nabla^{\alpha}\phi)(\nabla^{\beta}\phi)-\frac{1}{2}\;g_{\mu\nu}\;R[\nabla]_{\alpha\beta}\;(\nabla^{\alpha}\phi)(\nabla^{\beta}\phi)+R[\nabla]_{\mu\alpha\nu\beta}\;(\nabla^{\alpha}\phi)(\nabla^{\beta}\phi)-\frac{1}{2}\;g_{\mu\nu}\;(\nabla_{\beta}\nabla_{\alpha}\phi)(\nabla^{\beta}\nabla^{\alpha}\phi)+G[\nabla]_{\nu\alpha}\;(\nabla^{\alpha}\phi)(\nabla_{\mu}\phi)+G[\nabla]_{\mu\alpha}\;(\nabla^{\alpha}\phi)(\nabla_{\nu}\phi)+\frac{1}{2}\;R[\nabla]\,(\nabla_{\mu}\phi)(\nabla_{\nu}\phi)+(\nabla_{\mu}\nabla^{\alpha}\phi)(\nabla_{\nu}\nabla_{\alpha}\phi)-(\nabla_{\alpha}\nabla^{\alpha}\phi)(\nabla_{\nu}\nabla_{\mu}\phi)$$

Electromagnetic Field

In[ ]:=
```
DefTensor[A[-μ], M4];
DefTensor[F[-μ, -ν], M4, Antisymmetric[{1, 2}]]
```

```
** DefTensor: Defining tensor A[-μ].
** DefTensor: Defining tensor F[-μ, -ν].
```

Construct rules to compute $F = \mathrm{dA}$

In[ ]:=
```
FtodA = MakeRule[{F[-μ, -ν], PD[-μ][A[-ν]] - PD[-ν][A[-μ]]},
   MetricOn → All, ContractMetrics → True, UseSymmetries → True];
FtoDA = MakeRule[{F[-μ, -ν], CD[-μ][A[-ν]] - CD[-ν][A[-μ]]},
   MetricOn → All, ContractMetrics → True, UseSymmetries → True];
```

The rules can be used for any indices, upstairs or downstairs:

In[ ]:=
```
{F[-μ, -ν] /. FtodA, F[-μ, -ν] /. FtoDA, F[μ, -ν] /. FtoDA, F[μ, ν] /. FtoDA}
```

Out[ ]= $\{\partial_{\mu}A_{\nu}-\partial_{\nu}A_{\mu},\; \nabla_{\mu}A_{\nu}-\nabla_{\nu}A_{\mu},\; \nabla^{\mu}A_{\nu}-\nabla_{\nu}A^{\mu},\; \nabla^{\mu}A^{\nu}-\nabla^{\nu}A^{\mu}\}$

The stress energy tensor: vary with respect to $g^{\mu\nu}$.

In[ ]:=
```
Print[
 "T_{μν}= -2/Sqrt[-g] δ/δg^{μν} ( Sqrt[-g] (-1/4 F^{μν}F_{μν}))= ",
 tmn = (VarL[g[ μ,  ν], CD][ 1/2 F[-μ, -ν] F[μ, ν]] // FullSimplification[] //
     ScreenDollarIndices), "\n",
 "    = ", tmn /. FtodA // FullSimplification[] // ScreenDollarIndices, "\n",
 "    = ", tmn /. FtoDA // FullSimplification[] // ScreenDollarIndices
]
```

$$T_{\mu\nu}= -\frac{2}{\sqrt{-g}}\frac{\delta}{\delta g^{\mu\nu}}(\sqrt{-g}\ (-\frac{1}{4}F^{\mu\nu}F_{\mu\nu}))= F_{\mu}{}^{\alpha}\ F_{\nu\alpha}-\frac{1}{4}F_{\alpha\beta}\ F^{\alpha\beta}\ g_{\mu\nu}$$

$$= \partial_\alpha A_\nu\ \partial^\alpha A_\mu+\frac{1}{2}g^{\alpha\delta}\ g^{\beta\gamma}\ g_{\mu\nu}\ \partial_\beta A_\alpha\ \partial_\delta A_\gamma - \frac{1}{2}g^{\alpha\gamma}\ g^{\beta\delta}\ g_{\mu\nu}\ \partial_\beta A_\alpha\ \partial_\delta A_\gamma - \partial^\alpha A_\nu\ \partial_\mu A_\alpha - \partial^\alpha A_\mu\ \partial_\nu A_\alpha + g^{\alpha\beta}\ \partial_\mu A_\alpha\ \partial_\nu A_\beta$$

$$= (\nabla_\alpha A_\nu)(\nabla^\alpha A_\mu)+\frac{1}{2}g_{\mu\nu}\ (\nabla_\alpha A_\beta)(\nabla^\beta A^\alpha) - \frac{1}{2}g_{\mu\nu}\ (\nabla_\beta A_\alpha)(\nabla^\beta A^\alpha)-(\nabla^\alpha A_\nu)(\nabla_\mu A_\alpha)-(\nabla^\alpha A_\mu)(\nabla_\nu A_\alpha)+(\nabla_\mu A^\alpha)(\nabla_\nu A_\alpha)$$

Equations of motion: vary with respect to A_μ

In[ ]:=
```
Print[
 "δ/δA_μ(-1/4 F_{μν}F^{μν})= ", VarD[A[-μ], PD][-1/4 F[-μ, -ν] F[μ, ν] /. FtodA] // FullSimplification[] //
  ScreenDollarIndices, "\n",
 "                 = ",
 VarD[A[-μ], CD][-1/4 F[-μ, -ν] F[μ, ν] /. FtoDA] // FullSimplification[] //
  ScreenDollarIndices
]
```

$$\frac{\delta}{\delta A_\mu}(-\frac{1}{4}F_{\mu\nu}F^{\mu\nu})= -g^{\beta\gamma}\ g^{\mu\alpha}\ \partial_\gamma\partial_\alpha A_\beta + g^{\beta\gamma}\ g^{\mu\alpha}\ \partial_\gamma\partial_\beta A_\alpha + g^{\alpha\delta}\ g^{\beta\delta1}\ g^{\mu\gamma}\ \partial_\beta A_\alpha\ \partial_\delta g_{\gamma\delta1} - g^{\beta\gamma}\ g^{\delta\delta1}\ g^{\mu\alpha}\ \partial_\beta A_\alpha\ \partial_{\delta1} g_{\gamma\delta} + g^{\alpha\gamma}\ g^{\delta\delta1}\ g^{\mu\beta}\ \partial_\beta A_\alpha\ \partial_{\delta1} g_{\gamma\delta} - g^{\alpha\delta}\ g^{\beta\delta1}\ g^{\mu\gamma}\ \partial_\beta A_\alpha\ \partial_{\delta1} g_{\gamma\delta}$$

$$= -A^\alpha\ R[\nabla]^\mu{}_\alpha + \nabla_\alpha\nabla^\alpha A^\mu - \nabla^\mu\nabla_\alpha A^\alpha$$

Acknowledgements

This notebook has been programmed by Konstantinos Anagnostopoulos, Physics Department, National Technical University of Athens, Greece, while he was an instructor of the 4th year

undergraduate course "General Relativity and Cosmology". It was created for fun, but it may turn out to be useful to everyone studying the General Theory of Relativity for the first time.

Email: konstant@mail.ntua.gr

Web: http://physics.ntua.gr/konstant